Universidad del País Vasco — Euskal Herriko Unibertsitatea

eman ta zabal zazu

Autoevaluación

OCW 2018: *Utilizando Mathematica como apoyo al cálculo algebraico en los grados de Ingeniería*

Ejercicios propuestos resueltos (2 de 2)

Equipo docente del curso
Martín Yagüe, Luis
Unzueta Inchaurbe, Aitziber
Arrospide Zabala, Eneko
García Ramiro, María Begoña
Soto Merino, Juan Carlos
Alonso González, Erik

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I

OCW OpenCourseWare

AUTOEVALUACIÓN. EJERCICIOS (2 de 2)

Ejercicio nº1

Enunciado

Se consideran las expresiones: $E_1 = \dfrac{b^2 + ab}{a + ab}$ y $E_2 = \dfrac{(2 + 2a + b + ab)(a - 1)^2}{(a^3 - a^2 - a + 1)(b^2 + 3b + 2)}$

- a) Evalúe E_1, con cuatro cifras decimales, cuando $b = 17$ y $a = \sqrt{2}$.
- b) Defina una función $f(x)$ a partir de E_1 realizando las siguientes asignaciones: $b = x^2 - 1$ y $a = x^2 + 1$.
- c) Evalúe E_2, con cuatro cifras decimales, cuando $b = \pi$.
- d) Defina una función $g(x)$ a partir de E_2 realizando la siguiente asignación: $b = x^2$.
- e) Calcule los puntos de corte entre las gráficas de ambas funciones, $f(x)$ y $g(x)$.
- f) Obtenga el área del dominio limitado por las gráficas de las dos funciones.

Resolución

- a) Evalúe E_1, con cuatro cifras decimales, cuando $b = 17$ y $a = \sqrt{2}$.

```
e1 = (b^2 + a*b) / (a + a*b);
```

```
N[e1 /. {b → 17, a → Sqrt[2]}, 6]
```

```
12.2974
```

- b) Defina una función $f(x)$ a partir de E_1 realizando las siguientes asignaciones: $b = x^2 - 1$ y $a = x^2 + 1$.

```
f[x_] = e1 /. {b → x^2 - 1, a → x^2 + 1} // Simplify
```

$$\frac{2(-1 + x^2)}{1 + x^2}$$

- c) Evalúe E_2, con cuatro cifras decimales, cuando $b = \pi$.

$$\texttt{e2} = \frac{(2 + 2a + b + a*b)(a - 1)^2}{(a^3 - a^2 - a + 1)(b^2 + 3b + 2)} \texttt{ // Simplify}$$

$$\frac{1}{1 + b}$$

```
N[e2 /. b → Pi, 4]
```

```
0.2415
```

- d) Defina una función $g(x)$ a partir de E_2 realizando la siguiente asignación: $b = x^2$.

```
g[x_] = e2 /. b → x^2 // Simplify
```

$$\frac{1}{1 + x^2}$$

- e) Calcule los puntos de corte entre las gráficas de ambas funciones, $f(x)$ y $g(x)$.

```
s = NSolve[f[x] == g[x], x]
```

```
{{x → 1.22474}, {x → -1.22474}}
```

```
yc = {f[x] /. s[[1]], f[x] /. s[[2]]}
```

```
{0.4, 0.4}
```

Puntos de corte
P_1 (−1.22474, 0.40)
P_2 (1.22474, 0.40)

```
pts = Table[{x /. s[[i]], yc[[i]]}, {i, Length[s]}]
```

```
{{1.22474, 0.4}, {-1.22474, 0.4}}
```

```
pts1 = Graphics[{Green, PointSize[0.02], Point[pts[[1]]], Point[pts[[2]]]}, Axes → True, ImageSize → Small];
```

```
gq2 = Plot[{f[x], g[x]}, {x, x /. s[[1]], x /. s[[2]]},
   PlotLabels → Placed[{"f(x) = 2 (x^2 - 1)/(1 + x^2)", "g(x) = 1/(1 + x^2)"}, {{0.9, -1.3}, {{1, 0.8}}}],
   Filling → {1 → {{2}, LightGreen}}, Ticks → {{0, pts[[2, 1]], pts[[1, 1]]}, {0, pts[[1, 2]]}},
   Axes → True, AxesLabel → {"X", "Y"}, AspectRatio → Automatic];
```

```
Show[gq2, pts1]
```



- f) Obtenga el área del dominio limitado por las gráficas de las dos funciones.

```
area = 2 * Integrate[g[x] - f[x], {x, 0, pts[[1, 1]]}] (* simetría del dominio respecto al eje Oy *)
```

```
3.96179
```

Área del dominio
3.96179 u^2

Ejercicio nº2

Enunciado

Sea la matriz $A = \begin{pmatrix} 1 & -2 & 1 & 0 & 2 \\ 0 & 1 & 2 & -1 & 1 \\ 1 & 2 & 0 & -2 & 1 \\ 2 & 1 & 3 & -3 & 4 \\ 3 & -2 & 2 & -2 & 5 \end{pmatrix} \in \mathbb{M}_{5\times5}(\mathbb{R})$.

- a) Calcule su rango, $r(A)$.
- b) Obtenga una matriz, AR, equivalente por filas a A.
- c) Calcule la matriz $C = A + AR^{\mathrm{T}}$.
- d) ¿Es C invertible? En caso afirmativo, hallar C^{-1}.
- e) Resolver la ecuación: $X \cdot C + A = [0]_{5\times5}$.

Resolución

- Introducción de la matriz 3×3

```
Clear[A]; A = {{1, -2, 1, 0, 2}, {0, 1, 2, -1, 1}, {1, 2, 0, -2, 1}, {2, 1, 3, -3, 4}, {3, -2, 2, -2, 5}};
```

- a) Calcule su rango, $r(A)$.

```
MatrixRank[A]
```

3

- Solución: $\boxed{r(A) = 3}$

- b) Obtenga una matriz, AR, equivalente por filas a A.

```
AR = RowReduce[A]; MatrixForm[AR]
```

$$\begin{pmatrix} 1 & 0 & 0 & -\frac{8}{9} & \frac{11}{9} \\ 0 & 1 & 0 & -\frac{5}{9} & -\frac{1}{9} \\ 0 & 0 & 1 & -\frac{2}{9} & \frac{5}{9} \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

- c) Calcule la matriz $C = A + AR^{\mathrm{T}}$.

```
c = A + Transpose[AR]; MatrixForm[c]
```

$$\begin{pmatrix} 2 & -2 & 1 & 0 & 2 \\ 0 & 2 & 2 & -1 & 1 \\ 1 & 2 & 1 & -2 & 1 \\ \frac{10}{9} & \frac{4}{9} & \frac{25}{9} & -3 & 4 \\ \frac{38}{9} & -\frac{19}{9} & \frac{23}{9} & -2 & 5 \end{pmatrix}$$

- d) ¿Es C invertible? En caso afirmativo, hallar C^{-1}.

```
Det[c] (* det(c)≠0: matriz invertible *)
```

$$\frac{59}{9}$$

```
ci = Inverse[c]; MatrixForm[ci]
```

$$\begin{pmatrix} \frac{32}{59} & -\frac{1}{59} & \frac{33}{59} & -\frac{19}{59} & -\frac{4}{59} \\ -\frac{206}{59} & \frac{47}{59} & -\frac{76}{59} & -\frac{51}{59} & \frac{129}{59} \\ \frac{261}{59} & -\frac{10}{59} & \frac{94}{59} & \frac{46}{59} & -\frac{158}{59} \\ -\frac{229}{59} & \frac{68}{59} & -\frac{120}{59} & -\frac{65}{59} & \frac{154}{59} \\ -\frac{339}{59} & \frac{53}{59} & -\frac{156}{59} & -\frac{55}{59} & \frac{212}{59} \end{pmatrix}$$

- e) Resolver la ecuación: $X \cdot C + A = [0]_{5\times 5}$.
  - Aplicando propiedades de las matrices: $X = -AC^{-1}$

```
X = -A.ci; MatrixForm[X]
```

$$\begin{pmatrix} -\frac{27}{59} & -\frac{1}{59} & \frac{33}{59} & -\frac{19}{59} & -\frac{4}{59} \\ -\frac{206}{59} & -\frac{12}{59} & -\frac{76}{59} & -\frac{51}{59} & \frac{129}{59} \\ \frac{261}{59} & -\frac{10}{59} & \frac{35}{59} & \frac{46}{59} & -\frac{158}{59} \\ \frac{28}{59} & -\frac{23}{59} & -\frac{8}{59} & -\frac{24}{59} & -\frac{33}{59} \\ \frac{207}{59} & -\frac{12}{59} & \frac{101}{59} & \frac{8}{59} & -\frac{166}{59} \end{pmatrix}$$

Ejercicio nº3

Enunciado

Clasifique y resuelva el siguiente sistema de ecuaciones lineales: $\begin{pmatrix} 2 & 0 & 1 & 3 & 1 \\ 1 & 1 & 1 & 1 & 1 \\ 1 & 3 & -1 & -2 & 3 \\ -1 & 2 & 2 & 1 & 2 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \\ z \\ t \\ w \end{pmatrix} = \begin{pmatrix} -1 \\ 0 \\ 1 \\ 2 \end{pmatrix}$

Resolución

```
Clear[A, B, X]
```

- Definición de la matriz de coeficientes, A, la matriz columna de incógnitas, X, y la matriz columna de los términos independientes B:

```
A = {{2, 0, 1, 3, 1}, {1, 1, 1, 1, 1}, {1, 3, -1, -2, 3}, {-1, 2, 2, 1, 2}};
B = {-1, 0, 1, 2};
X = {x, y, z, t, w};
```

- El sistema no presenta parámetros reales; se resuelve con la función **Solve**:

```
Solve[A.X == B, {x, y, z, t, w}]
```

Solve: Equations may not give solutions for all "solve" variables.

$$\left\{\left\{y \to -3 - 5x,\ z \to \frac{9}{4} + \frac{13x}{4},\ t \to -2 - 3x,\ w \to \frac{11}{4} + \frac{15x}{4}\right\}\right\}$$

- Solución: $\boxed{\begin{cases} y = -3 - 5x \\ z = \frac{9}{4} + \frac{13}{4}x \\ t = -2 - 3x \\ w = \frac{11}{4} + \frac{15}{4}x \end{cases} \quad \forall x \in \mathbb{R}}$

- El sistema tiene infinitas soluciones en función de un parámetro: se trata de un sistema compatible indeterminado, siendo AM la matriz ampliada del sistema y n el número de incógnitas:

$$|A| \neq 0 \Rightarrow r(A) = r(AM) = 4 < 5 = n$$

```
AM = {{2, 0, 1, 3, 1, -1}, {1, 1, 1, 1, 1, 0}, {1, 3, -1, -2, 3, 1}, {-1, 2, 2, 1, 2, 2}};
```

```
{MatrixRank[A], MatrixRank[AM]}
```

```
{4, 4}
```

Ejercicio nº4

Enunciado

Sea el siguiente sistema de ecuaciones lineales: $\begin{cases} a\,x - 2\,y + 2\,z = 1 \\ 3\,x + 4\,a\,z = 1 \\ x + y + z = 0 \end{cases}$

- a) Clasifique, en función del parámetro $a \in \mathbb{R}$, el sistema de ecuaciones usando el rango de la matriz de coeficientes, A, y el de la ampliada, AM.
- b) Resuelva el sistema de ecuaciones en función del parámetro $a \in \mathbb{R}$.
- c) Interprete geométricamente la solución cuando el sistema sea compatible indeterminado.

Resolución

- a) Clasifique, en función del parámetro $a \in \mathbb{R}$, el sistema de ecuaciones usando el rango de la matriz de coeficientes, A, y el de la ampliada, AM.
  - Se definen la matriz de coeficientes, A, y la matriz ampliada, AM

```
Clear[A, AM, a];
A = {{a, -2, 2}, {3, 0, 4 a}, {1, 1, 1}};
AM = {{a, -2, 2, 1}, {3, 0, 4 a, 1}, {1, 1, 1, 0}};
```

  - Se calculan los valores $a \in \mathbb{R}$ que anulan el determinante de la matriz de coeficientes:

```
Solve[Det[A] == 0, a]
```

```
{{a → -3}, {a → 1}}
```

  - Caso 1: $a \neq -3 \wedge a \neq 1$ (en este caso: $|A| \neq 0 \Rightarrow r(A) = r(AM) = 3 = n$)
  - Caso 2: $a = -3$

```
{MatrixRank[A /. a → -3], MatrixRank[AM /. a → -3]}
```

```
{2, 3}
```

  - Caso 3: $a = 1$

```
{MatrixRank[A /. a → 1], MatrixRank[AM /. a → 1]}
```

```
{2, 2}
```

Parámetros	***Naturaleza***	***Rangos***
$a \neq -3 \wedge a \neq 1$	Compatible determinado	$r(A) = r(AM) = 3 = n$
$a = -3$	Incompatible	$r(A) = 2 \neq r(AM) = 3$
$a = 1$	Compatible indeterminado	$r(A) = r(AM) = 2 < n$

- b) Resuelva el sistema de ecuaciones en función del parámetro $a \in \mathbb{R}$.
  - Se utiliza la función **Reduce** para resolver el sistema

```
Reduce[A.{x, y, z} == {1, 1, 0}, {x, y, z}]
```

$$\left(a == 1 \,\&\&\, y == \frac{1}{4}(-1-x) \,\&\&\, z == \frac{1}{4}(1-3x)\right) \,||$$
$$\left((-1+a)(3+a) \neq 0 \,\&\&\, x == \frac{1}{3+a} \,\&\&\, y == \frac{1}{4}(-1-2x+ax) \,\&\&\, z == \frac{1}{4}(1-2x-ax)\right)$$

  - Caso 1: $a \neq -3 \wedge a \neq 1$

```
1/4 (-1 - 2 x + a x) /. x -> 1/(3 + a) // Simplify
```

$$-\frac{5}{4(3+a)}$$

```
1/4 (1 - 2 x - a x) /. x -> 1/(3 + a) // Simplify
```

$$\frac{1}{12+4a}$$

Parámetros	***Naturaleza***	***Solución***
$a \neq -3 \wedge a \neq 1$	Compatible determinado	$\begin{cases} x = \frac{1}{3+a} \\ y = -\frac{5}{4(3+a)} \\ z = \frac{1}{12+4a} \end{cases}$
$a = -3$	Incompatible	$--$
$a = 1$	Compatible indeterminado	$\begin{cases} y = \frac{1}{4}(-1-x) \\ z = \frac{1}{4}(1-3x) \end{cases} \quad \forall x \in \mathbb{R}$

- c) Interprete geométricamente la solución cuando el sistema sea compatible indeterminado.
  - Cada ecuación del sistema se interpreta geométricamente como un plano; las infinitas soluciones se corresponden con los puntos de la recta de intersección entre los tres planos

```
Clear[B]

B = {1, 1, 0}; prod = A.{x, y, z}; a = 1;

sist = Table[prod[[i]] == B[[i]], {i, Length[B]}]
```

$\{x - 2y + 2z == 1,\ 3x + 4z == 1,\ x + y + z == 0\}$

  - Representación gráfica:

```
p1 = InfinitePlane[{{1, 1, 1}, {0, 1, 3/2}, {3, 0, -1}}];

p2 = InfinitePlane[{{0, 1, 1/4}, {1, 1, -1/2}, {2, 0, -5/4}}];

p3 = InfinitePlane[{{0, -1, 1}, {1, -1, 0}, {3, 0, -3}}];

ls = Line[{{0, -1/4, 1/4}, {3, -1, -2}}];
```

```
Graphics3D[{LightGreen, p1, p2, p3, {Thickness[0.03], Red, ls}}, PlotRange → {{0, 3}, {-2, 1}, {-3, 2}},
 Boxed → True, Ticks → {{0, 3}, {-1, 0, 1}, {-2, 0, 1}}, Axes → True, AxesLabel → {"X", "Y", "Z"}]
```



Ejercicio nº5

Enunciado

Se consideran los subespacios vectoriales en el espacio vectorial $\mathbb{R}^3$:

$$S = \left\{ \begin{pmatrix} x \\ y \\ z \end{pmatrix} \in \mathbb{R}^3 \middle/ \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \lambda \begin{pmatrix} 1 \\ -2 \\ 0 \end{pmatrix} + \mu \begin{pmatrix} -1 \\ 2 \\ 2 \end{pmatrix} \lambda, \mu \in \mathbb{R} \right\}$$

$$T = \left\{ \begin{pmatrix} x \\ y \\ z \end{pmatrix} \in \mathbb{R}^3 \middle/ \begin{pmatrix} -1 & -2 & 3 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \\ z \end{pmatrix} = 0 \right\}$$

- a) Indique, razonadamente, una base y la dimensión de cada uno de los subespacios.
- b) Exprese S y T en forma cartesiana.
- c) Determine el subespacio vectorial $S \cap T$, obtenga una base y su dimensión.
- d) Interprete el resultado obtenido en el apartado anterior.

Resolución

- a) Indique, razonadamente, una base y la dimensión de cada uno de los subespacios.
  - El subespacio S está generado por la familia de vectores : $G_S = \{(1, -2, 0), (-1, 2, 2)\}$
  - Se estudia si la familia de vectores G_S es libre

```
Clear[A, B]; B = {s1 = {1, -2, 0}, s2 = {-1, 2, 2}}; MatrixRank[B]
```

2

  - La familia G_S, por tanto, es base de S: $\boxed{B_s = \{(1, -2, 0), (-1, 2, 2)\} \Rightarrow dim\, S = 2}$
  - Para obtener una base del subespacio vectorial T se obtiene su ecuación implícita:

```
Solve[ {-1, -2, 3}.{x, y, z} == 0, {y, z, x}]
```

··· Solve: Equations may not give solutions for all "solve" variables.

{{x → -2 y + 3 z}}

- La expresión general de un vector perteneciente al subespacio T es:

```
t = {x, y, z}; t = t /. {x → -2 y + 3 z}
```

```
{-2 y + 3 z, y, z}
```

- Siendo un sistema generador $G_T = \{(-2, 1, 0), (3, 0, 1)\}$

```
t1 = t /. {y → 1, z → 0}
t2 = t /. {y → 0, z → 1}
```

```
{-2, 1, 0}

{3, 0, 1}
```

- G_T es un sistema libre ya que $r(G_T) = 2$

```
MatrixRank[{t1, t2}]
```

```
2
```

- La familia G_T es, por tanto, base de T: $\boxed{B_T = \{(-2, 1, 0), (3, 0, 1)\} \Rightarrow dim\, T = 2}$

- b) Exprese S y T en forma cartesiana.
  - Obtención de la expresión en forma cartesiana del subespacio S:
    - Todo vector de S puede expresarse como combinación lineal de los vectores de una base cualquiera, en este caso: $B_s = \{(1, -2, 0), (-1, 2, 2)\}$
    - Entonces, $\forall\ (x, y, z) \in S$ el sistema de vectores $\{(1, -2, 0), (-1, 2, 2), (x, y, z)\}$ es ligado:

```
Solve[Det[{s1, s2, {x, y, z}}] == 0]
```

```
{{y → -2 x}}
```

    - Solución: $\boxed{S = \left\{(x, y, z) \in \mathbb{R}^3 \big/ 2\,x + y = 0\right\}}$
  - Obtención de la expresión en forma cartesiana del subespacio T:
    - En el apartado anterior se dedujo la ecuación implícita del subespacio T (es decir, la relación que deben verificar las coordenadas de un vector de T)
    - Utilizando dicha ecuación se puede deducir la forma cartesiana del subespacio T
    - Solución: $\boxed{T = \left\{(x, y, z) \in \mathbb{R}^3 \big/ x + 2\,y - 3\,z = 0\right\}}$

- c) Determine el subespacio vectorial $S \cap T$, obtenga una base y su dimensión.
  - En la intersección se encuentran los vectores que pertenecen simultáneamente a los dos subespacios
  - Verifican, por tanto, las ecuaciones de S y T: $S \cap T = \{(x, y, z)\ /\ 2\,x + y = 0,\ x + 2\,y - 3\,z = 0\}$
  - Entonces, $\forall\,(x, y, z) \in S \cap T$:

```
Solve[{2 x + y == 0, x + 2 y - 3 z == 0}, {x, y, z}]
```

```
Solve: Equations may not give solutions for all "solve" variables.
{{y → -2 x, z → -x}}
```

```
st = {x, y, z} /. {y → -2 x, z → -x} (* expresión genérica *)
```

```
{x, -2 x, -x}
```

```
st /. {x -> 1} (* base del subespacio *)
```

```
{1, -2, -1}
```

- Solución:
  - Subespacio: $\boxed{S \cap T = \{(x,\ -2x,\ -x)\ /\ \forall\ x \in \mathbb{R}\}}$
  - Base $\boxed{B_{S \cap T} = \{(1,\ -2,\ -1)\} \Rightarrow dim(S \cap T) = 1}$

- d) Interprete el resultado obtenido en el apartado anterior.
  - Interpretación geométrica:
    - El subespacio S es el plano de ecuación cartesiana: $2x + y = 0$
    - El subespacio T es el plano de ecuación cartesiana: $x + 2y - 3z = 0$
    - El subespacio $S \cap T$ es la recta de intersección entre ambos planos, su vector director es el obtenido en la base calculada: $\vec{d} = (1,\ -2,\ -1)$

```
p1 = InfinitePlane[{{1, -2, 0}, {0, 0, 1}, {0, 0, 0}}]; (* plano que contiene tres puntos *)
p2 = InfinitePlane[{{1, -2, -1}, {3, 0, 1}, {0, 0, 0}}]; (* plano que contiene tres puntos *)
r2 = Line[{{-1, 2, 1}, {1, -2, -1}}];
Graphics3D[{Opacity[0.35], LightGreen, p1, p2, Red, Thickness[0.015], r2},
 Boxed → False, Axes → True, AxesLabel → {"X", "Y", "Z"}, BoxRatios → Automatic, ImageSize → Small,
 Ticks → {{-1, 0, 1}, {-1, 0, 1}, {-1, 0, 1}}, PlotRange → {{-1, 1}, {-1, 1}, {-1, 1}}]
```



Ejercicio nº6

Enunciado

Sea la matriz: $A = \begin{pmatrix} 1 & 0 & 2 & -1 \\ 0 & 1 & 3 & 4 \\ 0 & 0 & 2 & n \\ 0 & 0 & 0 & 2 \end{pmatrix} \in \mathbb{M}_{4\times4}$

- a) Determine para qué valores de $n \in \mathbb{R}$ la matriz A es diagonalizable.
- b) Si la matriz A es diagonalizable calcule las matrices A y P tal que $D = P^{-1} \cdot A \cdot P$.

Resolución

- a) Determine para qué valores de $n \in \mathbb{R}$ la matriz A es diagonalizable.
  - Definición de la matriz y cálculo del polinomio característico

```
Clear[A, P, d, n]; A = {{1, 0, 2, -1}, {0, 1, 3, 4}, {0, 0, 2, n}, {0, 0, 0, 2}};

p[λ_] = CharacteristicPolynomial[A, λ] // Factor
```

$(-2 + \lambda)^2\ (-1 + \lambda)^2$

- Espectro de la matriz A : $\boxed{\sigma(A) = \{\lambda_1 = 2\ (k_1 = 2),\ \lambda_2 = 1\ (k_2 = 2)\}}$
- Dimensión del subespacio propio $V(2)$:

```
Reduce[(A - 2 IdentityMatrix[4]).{x, y, z, t} == {0, 0, 0, 0}, {x, y, z, t}]
```

$$\left(n == 0\ \&\&\ z == \frac{4x}{11} + \frac{y}{11}\ \&\&\ t == -\frac{3x}{11} + \frac{2y}{11}\right) \mid\mid \left(y == \frac{3x}{2}\ \&\&\ z == \frac{x}{2}\ \&\&\ t == 0\right)$$

  - Si $n = 0$: $V(2) = \left\{(x, y, z, t) \in \mathbb{R}^4 \big/ z = \frac{1}{11}(4x + y),\ t = -\frac{1}{11}(3x - 2y)\right\} \Rightarrow dim(V(2)) = 2$
  - Si $n \neq 0$: $V(2) = \left\{(x, y, z, t) \in \mathbb{R}^4 \big/ y = \frac{3}{2}x,\ z = \frac{1}{2}x,\ t = 0\right\} \Rightarrow dim(V(2)) = 1$

- Dimensión del subespacio propio $V(1)$:

```
Reduce[(A - IdentityMatrix[4]).{x, y, z, t} == {0, 0, 0, 0}, {x, y, z, t}]
```

```
z == 0 && t == 0
```

  - $\forall\, n \in \mathbb{R}$: $V(1) = \left\{(x, y, z, t) \in \mathbb{R}^4 \big/ z = 0,\ t = 0\right\} \Rightarrow dim(V(1)) = 2$

- Por lo tanto, la matriz sólo es diagonalizable si $n = 0$.
- Solución: $\boxed{A \text{ diagonalizable} \Longleftrightarrow n = 0}$
- <u>Nota</u>. No se puede utilizar el comando **Eigensystem[A]** cuando hay parámetros.

```
s = Eigensystem[A]
```

```
{{2, 2, 1, 1}, {{2, 3, 1, 0}, {0, 0, 0, 0}, {0, 1, 0, 0}, {1, 0, 0, 0}}}
```

  - La salida de la función induce a error ya que de su análisis se concluye que la matriz A no es diagonalizable independientemente del valor del parámetro n

- b) Si la matriz A es diagonalizable calcule las matrices A y P tal que $D = P^{-1} \cdot A \cdot P$
  - Se define la matriz A cuando $n = 0$:

```
A = {{1, 0, 2, -1}, {0, 1, 3, 4}, {0, 0, 2, n}, {0, 0, 0, 2}} /. n → 0; MatrixForm[A]
```

$$\begin{pmatrix} 1 & 0 & 2 & -1 \\ 0 & 1 & 3 & 4 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 2 \end{pmatrix}$$

```
s = Eigensystem[A]
```

```
{{2, 2, 1, 1}, {{-1, 4, 0, 1}, {2, 3, 1, 0}, {0, 1, 0, 0}, {1, 0, 0, 0}}}
```

```
d = DiagonalMatrix[s[[1]]]; MatrixForm[d]
```

$$\begin{pmatrix} 2 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
P = Transpose[s[[2]]]; MatrixForm[P]
```

$$\begin{pmatrix} -1 & 2 & 0 & 1 \\ 4 & 3 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix}$$

```
Inverse[P].A.P == d // Simplify
```

```
True
```

- Solución: $\boxed{P=\begin{pmatrix}-1&2&0&1\\4&3&1&0\\0&1&0&0\\1&0&0&0\end{pmatrix}}$ $\boxed{D=\begin{pmatrix}2&0&0&0\\0&2&0&0\\0&0&1&0\\0&0&0&1\end{pmatrix}}$

Ejercicio nº7

Enunciado

En el espacio vectorial $\mathbb{P}_2(x)$ de los polinomios en x de grado menor o igual que 2 se define el producto escalar siguiente:

$$\forall\ p(x),\ q(x)\in\mathbb{P}_2(x):\ \langle\ p(x),\ q(x)\ \rangle=\int_0^1 p(x)\cdot q(x)\,dx$$

- a) Determine el ángulo entre $p(x)=1$ y $q(x)=x$.
- b) ¿Para qué valores del parámetro $a\in\mathbb{R}$ son ortogonales $p_1(x)=x-a$ y $p_2(x)=x+a$?
- c) Calcule la matriz de Gram asociada a dicho producto escalar respecto a la base $B=\{1,\ x,\ x^2\}$.
- d) Calcule el subespacio ortogonal a $S=\mathcal{L}(\{1\})$.

Resolución

```
Clear[p, p1, p2, x, a, b]
```

- a) Determine el ángulo entre $p(x)=1$ y $q(x)=x$.
  - Introducción de los polinomios $p(x)=1$ y $q(x)=x$

```
p[x_] = 1; q[x_] = x;
```

  - Definición del producto escalar dado en el enunciado

```
pe[p_, q_] := Integrate[p[x] * q[x], {x, 0, 1}]
```

  - Cálculo del angulo entre los polinomios $p(x)$ y $q(x)$

```
ArcCos[pe[p, q] / (Sqrt[pe[p, p]] * Sqrt[pe[q, q]])]
```

$\frac{\pi}{6}$

  - Solución: $\boxed{\theta=\frac{\pi}{6}}$

- b) ¿Para qué valores del parámetro $a\in\mathbb{R}$ son ortogonales $p_1(x)=x-a$ y $p_2(x)=x+a$?
  - Se definen los polinomios $p_1(x)=x-a$ y $p_2(x)=x+a$

```
p1[x_] = x - a; p2[x_] = x + a;
```

  - Para que $p_1(x)$ y $p_2(x)$ sean ortogonales el producto escalar entre ambos debe ser nulo

```
Solve[pe[p1, p2] == 0]
```

$\left\{\left\{a\to-\frac{1}{\sqrt{3}}\right\},\left\{a\to\frac{1}{\sqrt{3}}\right\}\right\}$

  - Solución: $\boxed{p_1(x)\perp p_2(x)\Longleftrightarrow a=\frac{\sqrt{3}}{3}\ \vee\ a=-\frac{\sqrt{3}}{3}}$

- c) Calcule la matriz de Gram asociada a dicho producto escalar respecto a la base $B = \{1, x, x^2\}$
  - Se define $B = \{1, x, x^2\}$, base canónica de $\mathbb{P}_2(x)$:

```
b1[x_] = 1; b2[x_] = x; b3[x_] = x^2;
b = {b1, b2, b3};
```

  - Se calcula la matriz de Gram:

```
Table[pe[b[[i]], b[[j]]], {i, 1, Length[b]}, {j, 1, Length[b]}] // MatrixForm
```

$$\begin{pmatrix} 1 & \frac{1}{2} & \frac{1}{3} \\ \frac{1}{2} & \frac{1}{3} & \frac{1}{4} \\ \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \end{pmatrix}$$

  - Solución: $\boxed{G = \begin{pmatrix} 1 & \frac{1}{2} & \frac{1}{3} \\ \frac{1}{2} & \frac{1}{3} & \frac{1}{4} \\ \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \end{pmatrix}}$

- d) Calcule el subespacio ortogonal a $S = \mathcal{L}(\{1\})$.
  - Base del subespacio S: $B_S = \{1\}$, dado que $\{1\}$ no es el polinomio nulo
  - Subespacio ortogonal a S: $S^\perp = \{p(x) \in \mathbb{P}_2(x) / p(x) \perp S\}$
  - Se define un polinomio del espacio vectorial $\mathbb{P}_2(x)$ y se estudian las condiciones que deben satisfacer sus coeficientes para que sea ortogonal a la base B_S

```
pso[x_] = a1 x^2 + a2 x + a3; bs[x_] = 1;
Solve[pe[pso, bs] == 0]
```

$$\left\{\left\{a3 \to -\frac{a1}{3} - \frac{a2}{2}\right\}\right\}$$

```
pso[x_] = a1 x^2 + a2 x + a3 /. {a3 → -a1/3 - a2/2}
```

$$-\frac{a1}{3} - \frac{a2}{2} + a2\,x + a1\,x^2$$

  - Solución: $\boxed{S^\perp = \left\{p(x) = a_1\left(x^2 - \frac{1}{3}\right) + a_2\left(x - \frac{1}{2}\right) / a_1, a_2 \in \mathbb{R}\right\}}$
  - Sistema generador del subespacio: $F = \left\{x^2 - \frac{1}{3}, x - \frac{1}{2}\right\}$

```
pso1[x_] = pso[x] /. {a1 → 1, a2 → 0}; pso2[x_] = pso[x] /. {a1 → 0, a2 → 1};
F = {pso1[x], pso2[x]}
```

$$\left\{-\frac{1}{3} + x^2, -\frac{1}{2} + x\right\}$$

  - El sistema F es libre ya que $r(F) = 2$

```
M = CoefficientList[F, x, 3]
```

$$\left\{\left\{-\frac{1}{3}, 0, 1\right\}, \left\{-\frac{1}{2}, 1, 0\right\}\right\}$$

```
MatrixRank[M]
```

2

  - Base del subespacio $S^\perp$, $\boxed{B_{S^\perp} = \left\{x^2 - \frac{1}{3}, x - \frac{1}{2}\right\}}$